

Au fur et à mesure que l'on enrichit sa "programmothèque", la tentation d'utiliser des lambeaux de programmes originaux de fichiers disparates pour en faire un nouveau, devient de plus en plus pressante. L'idée est intéressante, mais comment la réaliser? Pour répondre à cette question, nous vous proposons un programme utilitaire taillé sur mesure pour le Junior Computer avec DOS, que tous les possesseurs d'un autre système pourront transplanter très facilement, à condition que le DOS (ou le BASIC) résident soit équipé d'un distributeur d'entrées/sorties permettant d'attaquer la mémoire comme un module périphérique, ainsi que le fait le Junior.

fusionner des fichiers BASIC

Un programme utilitaire qui permet de mettre bout à bout ou d'interclasser deux programmes distincts

La fonction du programme présenté ici est de fusionner ou de mettre bout à bout des programmes BASIC différents. Il est intéressant à ce titre, bien sûr, mais aussi parce qu'il fait appel à une propriété intéressante du DOS et du BASIC du Junior Computer; à savoir l'utilisation de la mémoire comme d'un organe d'entrée/sortie. Cette caractéristique est commune à la plupart des ordinateurs modernes.

Ce distributeur est un *commutateur logiciel* qui, lorsqu'il est programmé en conséquence, permet de mettre en relation la mémoire de travail (*workspace*) avec les organes périphériques conventionnels (clavier, écran, imprimante parallèle ou sérielle, etc) mais aussi, et c'est ce qui est passionnant ici, avec le reste de la mémoire. Dans le système DOS OS65D, le numéro de la mémoire en tant qu'organe d'E/S est 5. Pour les systèmes autres que celui du Junior Computer, nous vous prions de vous référer à votre manuel d'utilisation où vous trouverez les informations correspondantes, nécessaires à l'adaptation du programme. Le distributeur est géré par le DOS, mais

il est utilisable directement depuis le BASIC. De sorte que l'instruction LIST # 5, par exemple, provoque un transfert du fichier BASIC de la mémoire de travail (\$3A7E...) où il se trouve sous forme compactée (*tokenized*), vers l'adresse \$8000, à partir de laquelle on le retrouve en format ASCII intégral tel qu'il apparaît également sur l'écran ou sur une imprimante. L'adresse \$8000 est fixée automatiquement par le DOS, mais l'utilisateur peut la modifier à son gré.

Pour bien saisir l'intérêt de cette manipulation, il faut comprendre que dans la mémoire de travail de l'interpréteur, le fichier est compacté: les instructions du BASIC (mots réservés) y figurent sous une forme abrégée d'indicateurs (*token*) ou marqueurs, et non sous la forme d'une suite de codes ASCII correspondants aux lettres qui constituent le mot réservé du nom de l'instruction. Alors que dans la mémoire tampon, après l'exécution de l'instruction LIST # 5, on retrouve le fichier sous la forme qui nous est familière.

A l'inverse, le distributeur d'E/S nous permet de considérer la mémoire comme un

Tableau 1. Avec le listing ci-contre, vous êtes en possession d'un programme utilitaire qui vous permettra de fusionner deux fichiers BASIC (F\$ et S\$) après les avoir numérotés en conséquence avec la routine RSEQ du disque 2 du DOS d'Ohio Scientific modifié pour le Junior Computer (voir tableau 3). Signalons au passage que l'instruction de la ligne 2170 peut être remplacée par une instruction "DISK!" "MEM 8000;" (voir le manuel du DOS)...

Tableau 1.

```
2000 FORX=1TO24:PRINT:NEXT
2010 PRINTTAB(10) "-----"
2020 PRINTTAB(10) "-FILE MERGE UTILITY-"
2030 PRINTTAB(10) "-----"
2040 PRINT:PRINT:PRINTTAB(10) "written by A. Nachtmann"
2050 PRINT:PRINTTAB(10) "feb. 19, 1984"
2060 PRINT:PRINT:PRINT
2070 PRINT "Be sure that both files to be linked have different line numbers."
2080 PRINT "If both files have some common line numbers boot up your system"
2090 PRINT "with the RSEQ utility to renumber the lines."
2100 PRINT:INPUT "In which drive are the files to be merged A/B/C/D";D$
2110 D$=LEFT$(D$,1):D=ASC(D$):IF D<ASC("A") OR D>ASC("D") THEN2000
2120 PRINT:INPUT "enter first file name ";F$
2130 INPUT "enter second file name";S$
2140 PRINT:INPUT "are you ready";I$
2150 IF LEFT$(I$,1)<>"Y" THEN2140
2160 REM---RESET MEMORY INPUT POINTER
2170 POKE9098,0:POKE9099,128
2180 DISK!"SE A":DISK!"CA E400=12,7":DISK!"SE "+D$:DISK!"GO E481"
2190 A1=8*16^3+11: A2=8*16^3+2*16+4
2200 REM---
2210 A=A1
2220 FORX=1 TO LEN(F$)
2230 POKE A,ASC(MID$(F$,X,1)):A=A+1
2240 NEXT
2250 REM---
2260 A=A2
2270 FOR X=1TO LEN(S$)
2280 POKEA,ASC(MID$(S$,X,1)):A=A+1
2290 NEXT
2300 POKE8993,16
```

Tableau 2.

	0	1	2	3	4	5	6	7	8	9	A	B	C	D	E	F	
E400:	50	4F	4B	45	38	39	39	33	2C	31	0D	0A	00	0D	0A	44	POKE8993,1.....D
E410:	49	53	4B	21	22	4C	4F	20	20	20	20	20	20	22	3A	ISK!"LO	:"
E420:	4C	49	53	54	23	35	0D	0A	44	49	53	4B	21	22	4C	4F	LIST#5..DISK!"LO
E430:	20	20	20	20	20	20	20	22	3A	4C	49	53	54	23	35	0D	:"LIST#5.
E440:	0A	44	49	53	4B	21	22	47	4F	20	45	34	35	32	22	0D	.DISK!"GO E452".
E450:	0A	00	AE	91	23	AD	92	23	8E	66	E4	8D	67	E4	A2	00	
E460:	BD	00	E4	F0	1B	8D	FF	FF	EE	66	E4	D0	03	EE	67	E4	
E470:	AD	66	E4	8D	91	23	AD	67	E4	8D	92	23	E8	D0	E1	60	
E480:	60	A2	00	A9	80	8E	66	E4	8D	67	E4	A2	0D	D0	D1		

Tableau 3.

	0	1	2	3	4	5	6	7	8	9	A	B	C	D	E	F	
2200:	A9	01	8D	5E	26	20	BC	26	A9	2A	85	FF	20	54	27	86	...^& .&.*... T'
2210:	FE	20	67	29	A0	BF	20	EC	22	F0	03	88	D0	F8	8C	00	. g)
2220:	23	20	AC	15	20	9E	0F	20	94	15	20	73	2D	0D	0A	2D	# s-..
2230:	2D	20	44	49	53	4B	20	32	20	2D	2D	0D	0A	0A	00	4C	- DISK 2 --....L
2240:	E6	2A															

organe d'entrée, au lieu que ce soit le clavier. Le programme d'interclassement fait abondamment appel à ces possibilités.

BASIC en fusion

Le programme comporte une partie en langage machine (tableau 2) et une autre en BASIC par laquelle nous commencerons. Une fois qu'il connaît l'unité sur laquelle se trouvent les fichiers à fusionner (D\$) ainsi que leur nom (F\$ et S\$ sont deux noms qui doivent se trouver dans le répertoire de l'unité désignée par D\$) — lignes 2000...2160 — le processeur initialise le pointeur désignant l'adresse à partir de laquelle on retrouvera le fichier transféré en mémoire. Il charge ensuite un programme en langage machine et une table de consultation en \$E400 (depuis le secteur 7 de la piste 12; il reste en effet de la place derrière le répertoire!). Le programme en langage machine est lancé par l'instruction GO de la ligne 2180; il place dans le tampon (\$8000...) la série d'instructions en mode direct (sans numéros de ligne) que l'on trouve dans la partie droite du tableau 2. De la ligne 2190 à la ligne 2290, le programme BASIC s'occupe de placer le nom des fichiers à fusionner (F\$ et S\$) derrière les deux instructions LO en mode direct qu'il avait placées auparavant dans le tampon (voir ci-dessus). L'instruction de la ligne 2300 programme le distributeur d'entrée de telle sorte que l'organe d'entrée soit la mémoire. L'éditeur BASIC reçoit à présent la séquence d'instructions du tampon à partir de l'adresse \$8000, comme si elles étaient entrées une à une via le clavier... et il les exécute l'une après l'autre. C'est à dire qu'il charge le fichier F\$, le transfère en \$8000 (LIST # 5), qu'il charge ensuite le fichier S\$ et le transfère à son tour derrière le fichier F\$ au delà de \$8000. Puis il exécute l'instruction DISK! "GO E452", la dernière qu'il reçoive en mode direct de la mémoire en tant qu'organe d'entrée. Le programme en langage machine en \$E452 place derrière les deux fichiers chargés dans le tampon \$8000 l'instruction POKE 8993,1 en mode direct (elle n'est pas précédée par un numéro de ligne et sera donc exécutée aussitôt que l'interpré-

teur la rencontrera). La fonction de cette instruction finale est de rétablir le distributeur d'entrée dans sa configuration originale (le clavier redevient l'organe d'entrée. Mais nous n'en sommes pas encore là...). A présent, l'éditeur BASIC charge dans sa mémoire de travail les deux fichiers F\$ et S\$ qu'il reçoit du tampon \$8000, pour n'en faire qu'un seul, qu'il compacte et liste au fur et à mesure. Arrivé au terme de la dernière ligne numérotée du second fichier, il trouve l'instruction POKE 8993,1 qu'il exécute en mode direct; et c'est ainsi que l'organe d'entrée actif redevient le clavier. Donnez l'instruction LIST, et vous constaterez sur votre écran que la mémoire de travail contient bel et bien les deux fichiers F\$ et S\$.

RSEQ

Pour être en mesure de fusionner efficacement des programmes existants, il faut pouvoir manipuler aisément la numérotation des lignes de chacun des deux fichiers à interclasser, puis éventuellement celle du fichier unique résultant de la fusion. Il existe sur le disque 2 de la série des 5 disques fournis avec le DOS d'Ohio Scientific un programme utilitaire RSEQ que l'on peut mettre à contribution pour effectuer cette tâche. Dans nos publications antérieures, nous n'avions pas encore communiqué la procédure d'adaptation du disque 2 au Junior Computer. Avec le tableau 3, c'est chose faite; de sorte que nos lecteurs pourront désormais facilement et rapidement modifier la numérotation des lignes de leurs fichiers BASIC, notamment de ceux qu'ils désirent fusionner.

La procédure d'adaptation est fort simple. On commence par effectuer une copie de la disquette originale (il faut toujours assurer ses arrières!), puis on charge la piste 0 du disque 2 à l'aide de TRACK 0 R/W UTILITY (RA200) en \$A200 (ou ailleurs...). Il faut ensuite modifier le contenu de cette piste conformément au tableau 3, et remettre la première page de la piste 0 ainsi modifiée sur la disquette (WA200/2200,8). C'est tout!

fusionner des fichiers BASIC
elektor juin 1984

Tableau 2. Une partie de File Merge Utility est écrite en langage machine; elle assure la mise en place d'instructions en mode direct (sans numéro de ligne) dans la mémoire de travail de l'interpréteur BASIC.

Tableau 3. Modifications à apporter au contenu de la piste 0 du disque 2 d'Ohio Scientific pour pouvoir l'utiliser avec le Junior Computer.